Calculations of ghost cells for finite difference (FD) or finite volume (FV) schemes

---

Cubic extrapolation finite volume with Dirichlet BC

```
> restart:
> u:=x->d*x^3+a*x^2+b*x+c;
```

$$u := x \mapsto d \cdot x^3 + a \cdot x^2 + b \cdot x + c \tag{1}$$

```
> u_bar:=unapply(int(u(x),x=i*h..(i+1)*h)/h,i);
```

$$u\_bar := i \mapsto \frac{1}{h}\left(\frac{d \cdot \left((i+1)^4 \cdot h^4 - i^4 \cdot h^4\right)}{4} + \frac{a \cdot \left((i+1)^3 \cdot h^3 - i^3 \cdot h^3\right)}{3} + \frac{b \cdot \left((i+1)^2 \cdot h^2 - i^2 \cdot h^2\right)}{2} + c \cdot \left((i+1) \cdot h - i \cdot h\right)\right) \tag{2}$$

```
> sol:=solve({u(0)=u_BC, u_bar(0)=u1, u_bar(1)=u2, u_bar(2)=u3},{a,
  b,c,d});
```

$$sol := \left\{a = -\frac{10\,u1 - 5\,u2 + u3 - 6\,u\_BC}{2\,h^2},\ b = \frac{85\,u1 - 23\,u2 + 4\,u3 - 66\,u\_BC}{18\,h},\ c = u\_BC,\ d = \frac{11\,u1 - 7\,u2 + 2\,u3 - 6\,u\_BC}{9\,h^3}\right\} \tag{3}$$

```
> assign(sol);
> u0:=simplify(u_bar(-1));
```

$$u0 := -\frac{13\,u1}{3} + \frac{5\,u2}{3} - \frac{u3}{3} + 4\,u\_BC \tag{4}$$

```
> um1:=simplify(u_bar(-2));
```

$$um1 := -\frac{70\,u1}{3} + \frac{32\,u2}{3} - \frac{7\,u3}{3} + 16\,u\_BC \tag{5}$$

Upwind flux at i+1/2: 1/6*(-u(i-1)+5*u(i)+2*u(i+1));

```
> 1/6*(-um1+5*u0+2*u1); # Flux NOT exact on inflow face
```

$$\frac{11\,u1}{18} - \frac{7\,u2}{18} + \frac{u3}{9} + \frac{2\,u\_BC}{3} \tag{6}$$

```
> 1/6*(-u0+5*u1+2*u2);
```

$$\frac{5\,u1}{4} + \frac{u2}{4} - \frac{u\_BC}{2} \tag{7}$$

Quadratic extrapolation finite difference with Dirichlet BC

```
> restart:
> u:=x->a*x^2+b*x+c;
```

$$u := x \mapsto a \cdot x^2 + b \cdot x + c \tag{8}$$

```
> sol:=solve({u(0)=u_BC, u(h/2)=u1, u(3*h/2)=u2},{a,b,c});
```

$$sol := \left\{a = -\frac{2\,(3\,u1 - u2 - 2\,u\_BC)}{3\,h^2},\ b = \frac{9\,u1 - u2 - 8\,u\_BC}{3\,h},\ c = u\_BC\right\} \tag{9}$$

```
> assign(sol);
> u0:=u(-h/2);
```

$$u0 := -2\,u1 + \frac{u2}{3} + \frac{8\,u\_BC}{3} \tag{10}$$

```
> um1:=u(-3*h/2);
```

$$um1 := -9\,u1 + 2\,u2 + 8\,u\_BC \tag{11}$$

Quadratic finite volume extrapolation with Neumann BC

```
> restart:
> u:=x->a*x^2+b*x+c;
```

$$u := x \mapsto a\cdot x^2 + b\cdot x + c \tag{12}$$

```
> u_bar:=unapply(int(u(x),x=i*h..(i+1)*h)/h,i);
```

$$u\_bar := i \tag{13}$$

$$\mapsto \frac{\frac{a\cdot\left((i+1)^3\cdot h^3 - i^3\cdot h^3\right)}{3} + \frac{b\cdot\left((i+1)^2\cdot h^2 - i^2\cdot h^2\right)}{2} + c\cdot\left((i+1)\cdot h - i\cdot h\right)}{h}$$

```
> sol:=solve({eval(diff(u(x),x),x=0)=u_prime_BC, u_bar(0)=u1, u_bar
  (1)=u2},{a,b,c});
```

$$sol := \left\{a = -\frac{h\,u\_prime\_BC + u1 - u2}{2\,h^2},\ b = u\_prime\_BC,\ c = -\frac{h\,u\_prime\_BC}{3} + \frac{7\,u1}{6} - \frac{u2}{6}\right\} \tag{14}$$

```
> assign(sol);
> u0:=simplify(u_bar(-1)); # Same as linear extrapolation by
  symmetry
```

$$u0 := -h\,u\_prime\_BC + u1 \tag{15}$$

```
> um1:=simplify(u_bar(-2));
```

$$um1 := -3\,h\,u\_prime\_BC + u2 \tag{16}$$

Quadratic finite volume extrapolation with Dirichlet BC

```
> restart:
> u:=x->a*x^2+b*x+c;
```

$$u := x \mapsto a\cdot x^2 + b\cdot x + c \tag{17}$$

```
> u_bar:=unapply(int(u(x),x=i*h..(i+1)*h)/h,i);
```

$$u\_bar := i \tag{18}$$

$$\mapsto \frac{\frac{a\cdot\left((i+1)^3\cdot h^3 - i^3\cdot h^3\right)}{3} + \frac{b\cdot\left((i+1)^2\cdot h^2 - i^2\cdot h^2\right)}{2} + c\cdot\left((i+1)\cdot h - i\cdot h\right)}{h}$$

```
> sol:=solve({u(0)=u_BC, u_bar(0)=u1, u_bar(1)=u2},{a,b,c});
```

$$sol := \left\{a = -\frac{3\,(3\,u1 - u2 - 2\,u\_BC)}{4\,h^2},\ b = \frac{7\,u1 - u2 - 6\,u\_BC}{2\,h},\ c = u\_BC\right\} \tag{19}$$

```
> assign(sol);
> u0:=simplify(u_bar(-1));
```

$$u0 := -\frac{5\,u1}{2} + \frac{u2}{2} + 3\,u\_BC \tag{20}$$

```
> um1:=simplify(u_bar(-2));
```

$$um1 := -\frac{21\,u1}{2} + \frac{5\,u2}{2} + 9\,u\_BC \tag{21}$$

```
Upwind flux at i+1/2: 1/6*(-u(i-1)+5*u(i)+2*u(i+1));
> 1/6*(-um1+5*u0+2*u1); # This turns out to be exact!
```

$$u\_BC \tag{22}$$

```
> 1/6*(-u0+5*u1+2*u2);
```

$$\frac{5\,u1}{4} + \frac{u2}{4} - \frac{u\_BC}{2} \tag{23}$$

Quadratic extrapolation finite difference without BCs

```
> restart:
> u:=x->a*x^2+b*x+c;
```

$$u := x \mapsto a\cdot x^2 + b\cdot x + c \tag{24}$$

```
> sol:=solve({u(h/2)=u1, u(3*h/2)=u2, u(5*h/2)=u3},{a,b,c});
```

$$sol := \left\{a = \frac{u1 - 2\,u2 + u3}{2\,h^2},\ b = -\frac{2\,u1 - 3\,u2 + u3}{h},\ c = \frac{15\,u1}{8} - \frac{5\,u2}{4} + \frac{3\,u3}{8}\right\} \tag{25}$$

```
> assign(sol);
> u0:=u(-h/2);
```

$$u0 := 3\,u1 - 3\,u2 + u3 \tag{26}$$

```
> um1:=u(-3*h/2);
```

$$um1 := 6\,u1 - 8\,u2 + 3\,u3 \tag{27}$$

Quadratic extrapolation FV without BC

```
> restart:
> u:=x->a*x^2+b*x+c;
```

$$u := x \mapsto a\cdot x^2 + b\cdot x + c \tag{28}$$

```
> u_bar:=unapply(int(u(x),x=i*h..(i+1)*h)/h,i);
```

$$u\_bar := i \tag{29}$$

$$\mapsto \frac{\frac{a\cdot\left((i+1)^3\cdot h^3 - i^3\cdot h^3\right)}{3} + \frac{b\cdot\left((i+1)^2\cdot h^2 - i^2\cdot h^2\right)}{2} + c\cdot\left((i+1)\cdot h - i\cdot h\right)}{h}$$

```
> sol:=solve({u_bar(0)=u1, u_bar(1)=u2, u_bar(2)=u3},{a,b,c});
```

$$sol := \left\{a = \frac{u1 - 2\,u2 + u3}{2\,h^2},\ b = -\frac{2\,u1 - 3\,u2 + u3}{h},\ c = \frac{11\,u1}{6} - \frac{7\,u2}{6} + \frac{u3}{3}\right\} \quad (30)$$

```
> assign(sol);
> u0:=simplify(u_bar(-1));
```

$$u0 := 3\,u1 - 3\,u2 + u3 \quad (31)$$

```
> um1:=simplify(u_bar(-2));
```

$$um1 := 6\,u1 - 8\,u2 + 3\,u3 \quad (32)$$

```
> u(-h/2); # Different result (incorrect)
```

$$\frac{71\,u1}{24} - \frac{35\,u2}{12} + \frac{23\,u3}{24} \quad (33)$$

```
> eval(diff(u(x),x),x=0);
```

$$-\frac{2\,u1 - 3\,u2 + u3}{h} \quad (34)$$

```
================================================================
========
```

OLD calculations for diffusion accuracy with Dirichlet BCs

```
> restart:
> u:=x->a*x^2+b*x+c;
```

$$u := x \mapsto a \cdot x^2 + b \cdot x + c \quad (35)$$

```
> sol:=solve({u(0)=u_BC, u(h/2)=u1, u(3*h/2)=u2},{a,b,c});
```

$$sol := \left\{a = -\frac{2\,(3\,u1 - u2 - 2\,u\_BC)}{3\,h^2},\ b = \frac{9\,u1 - u2 - 8\,u\_BC}{3\,h},\ c = u\_BC\right\} \quad (36)$$

```
> assign(sol);
> u0:=u(-h/2);
```

$$u0 := -2\,u1 + \frac{u2}{3} + \frac{8\,u\_BC}{3} \quad (37)$$

```
> expand(simplify((u0-2*u1+u2)/h^2));
```

$$-\frac{4\,u1}{h^2} + \frac{4\,u2}{3\,h^2} + \frac{8\,u\_BC}{3\,h^2} \quad (38)$$

```
----
> restart:
> u0 := -3*u1 + u2 - u3/5 + (16*u_BC)/5; # Cubic extrapolation
```

$$u0 := -3\,u1 + u2 - \frac{u3}{5} + \frac{16\,u\_BC}{5} \quad (39)$$

```
> Laplacian_BC:=-collect(expand(simplify(d_1_2*(u1-u0)/h-d_3_2*(u2-
  u1)/h)/h),{u1,u2});
```

$$Laplacian\_BC := -\left(\frac{4\,d\_1\_2}{h^2} + \frac{d\_3\_2}{h^2}\right) u1 - \left(-\frac{d\_1\_2}{h^2} - \frac{d\_3\_2}{h^2}\right) u2 - \frac{d\_1\_2\,u3}{5\,h^2} + \frac{16\,d\_1\_2\,u\_BC}{5\,h^2} \quad (40)$$